PARABOLA

In the projectiviced affine euclidean and with respecto the orthonormal coordinate system R={O;B={e_1,e_2}} we consider the projective conic C of equation $x_1^2+x_2^2-2x_1x_2-2x_0x_1-2x_0x_2=0$.

Classify C

The matrix of the conic in R is:

```
> A:=matrix(3,3,[0,-1,-1,-1,1,-1,-1,-1,1]);
```

$$A := \begin{bmatrix} 0 & -1 & -1 \\ -1 & 1 & -1 \\ -1 & -1 & 1 \end{bmatrix} \qquad \textbf{(1.1)}$$

```
> with(linalg):
> A00:=submatrix(A,2..3,2..3);
```

$$A00 := \begin{bmatrix} 1 & -1 \\ -1 & 1 \end{bmatrix} \qquad \textbf{(1.2)}$$

```
> det(A00); det(A);
```

$$0$$

$$-4 \qquad \textbf{(1.3)}$$

Therefore C is a nondegenerate conic of parabolic type, it is a PARABOLA.

Let us draw the graph of the affine conic:

```
> Xafin:=matrix(3,1,[1,x[1],x[2]]):
> Cafin:=simplify(evalm(transpose(Xafin)&*A&*Xafin))[1,1];
```

$$Cafin := -2\,x_1 - 2\,x_2 + x_1^2 - 2\,x_1\,x_2 + x_2^2 \qquad \textbf{(1.4)}$$

```
> with(plots):
> implicitplot({Cafin},x[1]=-5..5,x[2]=-5..5,color=red,axes=
  normal);
```



Determine the center, the axis, the vertices and the asymptotes of C if there are any.

Parabolas verify:
- They have a proper center which is the pole of the line at infinity. It can also be computed as [det(A00), det(A01), det(A02)].
- The only have one axis which is the axis of symmetry of the conic. The axis is perpendicular to the line at infinity and intersectes the conic in its point at infinity (the center of the parabola)
and also in the vertex (its only intersection with the conic in a proper point). The axis is the polar line of the point at infinity corresponding to the eigenvalue of the nonzero eigenvector of A00.
- They do not have proper asymptotes. We say that the line at infinity is the only asymptote of the parabola.

```
> Center:=evalm(inverse(A)&*matrix(3,1,[1,0,0]));
```

Therefore the coordinates of the CENTER are [0,1,1].

$$Center := \begin{bmatrix} 0 \\ -\frac{1}{2} \\ -\frac{1}{2} \end{bmatrix} \tag{2.1}$$

```
> eigenvectors(A00);
```

$$\left[0, 1, \left\{\left[\begin{array}{cc} 1 & 1 \end{array}\right]\right\}\right], \left[2, 1, \left\{\left[\begin{array}{cc} -1 & 1 \end{array}\right]\right\}\right] \tag{2.2}$$

```
> X:=matrix(3,1,[x[0],x[1],x[2]]): AXIS:=evalm(matrix(1,3,[0,-1,
  1])&*A&*X)[1,1];
```

$$AXIS := -2\,x_1 + 2\,x_2 \tag{2.3}$$

Observe that the direction of the axis is the eigenvector associated to the eigenvalue 0, therefore its point at infinity is [0,1,1].

We compute the vertex, proper intersection of C with the axis:

```
> C:=simplify(evalm(transpose(X)&*A&*X))[1,1];
```

$$C := -2\,x_0\,x_1 - 2\,x_0\,x_2 + x_1^2 - 2\,x_1\,x_2 + x_2^2 \tag{2.4}$$

```
> solve({AXIS, C},{x[0],x[1],x[2]});
```

$$\{x_0 = 0, x_1 = x_2, x_2 = x_2\}, \{x_0 = x_0, x_1 = 0, x_2 = 0\} \tag{2.5}$$

Thus the VERTEX equals [1,0,0].

We draw the graph of the affine conic with the axis.

```
> GconicAffine:=implicitplot({Cafin},x[1]=-5..5,x[2]=-5..5,color=
  red,axes=normal):
> GaxisAffine:=implicitplot({AXIS},x[1]=-4..4,x[2]=-4..4,color=
  blue,axes=normal):
> display([GconicAffine,GaxisAffine]);
```



Find the tangent lines to C parallel to the line r with equation 2 $x_1 - x_2 = 0$.

The parallel lines to r have the same point at infinity Pinf than r. To be tangent to C they also have to contain one of the intersection points Q1 or Q2 of C with the polar line of Pinf.

```
> Pinf:=matrix(3,1,[0,1,2]);
```

$$Pinf := \begin{bmatrix} 0 \\ 1 \\ 2 \end{bmatrix} \tag{3.1}$$

```
> Polar:=evalm(transpose(Pinf)&*A&*X)[1,1];
```

$$Polar := -3\,x_0 - x_1 + x_2 \tag{3.2}$$

```
> solve({Polar, C},{x[0],x[1],x[2]});
```

$$\left\{x_0 = \frac{4}{15}\,x_2, x_1 = \frac{1}{5}\,x_2, x_2 = x_2\right\}, \{x_0 = 0, x_1 = x_2, x_2 = x_2\} \tag{3.3}$$

```
> Q1:=matrix(3,1,[0,1,1]); Q2:=matrix(3,1,[4, 3, 15]);
```

$$Q1 := \begin{bmatrix} 0 \\ 1 \\ 1 \end{bmatrix}$$

$$Q2 := \begin{bmatrix} 4 \\ 3 \\ 15 \end{bmatrix} \quad \textbf{(3.4)}$$

The tangents we are looking for are the polar line of Q2 and the line at infinity.

```
> PolarQ2:=evalm(transpose(Q2)&*A&*X)[1,1]=0;
```

$$PolarQ2 := -18\,x_0 - 16\,x_1 + 8\,x_2 = 0 \quad \textbf{(3.5)}$$

We draw the situation.

```
> GrectaAfin:=implicitplot({2*x[1]-x[2]=0},x[1]=-5..5,x[2]=-5..5,
  color=green,axes=normal):
> GpolarAfin:=implicitplot({-3-x[1]+x[2]=0},x[1]=-5..5,x[2]=-5.
  .5,color=blue,axes=normal):
> GPolarQ2Afin:=implicitplot({-18-16*x[1]+8*x[2]=0},x[1]=-5..5,x
  [2]=-5..5,color=orange,axes=normal):
> display(GconicaAfin,GrectaAfin,GpolarAfin,GPolarQ2Afin);
```




```
>
```

Find the poles of the line $x_1 + \lambda x_2 + x_0 = 0$ with respect to C.

```
> evalm(inverse(A)&*matrix(3,1,[1,1,lambda]));
```

$$\begin{bmatrix} -\frac{1}{2} - \frac{1}{2}\lambda \\ -\frac{1}{4} - \frac{1}{4}\lambda \\ -\frac{3}{4} + \frac{1}{4}\lambda \end{bmatrix} \quad (4.1)$$

```
> P1:=matrix(1,3,[-1/2,-1/4,-3/4]);P2:=matrix(1,3,[-1/2,-1/4,1/4]
  );
```

$$P1 := \begin{bmatrix} -\frac{1}{2} & -\frac{1}{4} & -\frac{3}{4} \end{bmatrix}$$

$$P2 := \left[\begin{array}{ccc} -\frac{1}{2} & -\frac{1}{4} & \frac{1}{4} \end{array}\right] \qquad (4.2)$$

The poles are in the projective line joining P1 and P2.

Invariants of the conic and reduced equation of the conic in terms of the invariants.

```
> det(A); det(A00); eigenvalues(A00); trace(A00);
```

$$-4$$
$$0$$
$$0, 2$$
$$2 \qquad (5.1)$$

```
> lambda[2]:=2;p:=sqrt(-det(A)/(lambda[2]^3));x[2]^2=2*p*x[1];
```

$$\lambda_2 := 2$$
$$p := \frac{1}{2}\sqrt{2}$$
$$x_2^2 = \sqrt{2}\,x_1 \qquad (5.2)$$

Determine a new coordinate system R' so that the equation of C in R' is the reduced equation of C. Give the matrix of the change of coordinates and check that we get reduced equation in R'.

The new coordinate system R'={O', B'} has:
- origin O' the vertex of the conic [0,0].
- basis B' an orthonormal basis given by the eigenvectors of A00.

Then the matrix of change of coordinates is:

```
> Q:=matrix(3,3,[1,0,0,0, 1/sqrt(2),-1/sqrt(2),0,1/sqrt(2),1/sqrt
  (2)]);
```

$$Q := \left[\begin{array}{ccc} 1 & 0 & 0 \\ 0 & \frac{1}{2}\sqrt{2} & -\frac{1}{2}\sqrt{2} \\ 0 & \frac{1}{2}\sqrt{2} & \frac{1}{2}\sqrt{2} \end{array}\right] \qquad (6.1)$$

The matrix of the conic in R' is:

```
> Ap:=evalm(transpose(Q)&*A&*Q);
```

$$Ap := \begin{bmatrix} 0 & -\sqrt{2} & 0 \\ -\sqrt{2} & 0 & 0 \\ 0 & 0 & 2 \end{bmatrix} \tag{6.2}$$

The equation of the affine conic in R' is:

```
> reduced:=simplify(evalm(transpose(Xafin)&*Ap&*Xafin))[1,1]=0;
```

$$reduced := -2\sqrt{2}\,x_1 + 2\,x_2^2 = 0 \tag{6.3}$$